

## A Rank 7 Pfaffian System on a 15-Dimensional Manifold with $\mathfrak{f}_4$ Symmetry Algebra

### Synopsis

- Let  $\Omega$  be a differential system on a manifold  $M$ . The infinitesimal symmetry algebra of  $\Omega$  is the set of all vectors fields  $X$  on  $M$  such that  $\mathcal{L}_X(\Omega) \subset \Omega$ .
- In this worksheet we present an example, due to E. Cartan [2] p. 418 (see also K. Yamaguchi [3] p. 482), of a rank 7 Pfaffian system on a 15-dimensional manifold whose infinitesimal symmetry algebra is the split real form of the exceptional Lie algebra  $\mathfrak{f}_4$ .
- The analysis of these symmetries presents a number of computational challenges which we address in Sections 1 and 2.
- Since the infinitesimal symmetry group  $\mathfrak{f}_4$  of this differential system acts transitively on the underlying manifold, this manifold is locally a homogeneous space  $F_4/H$ . In Section 4, we identify the Lie algebra of the subgroup  $H$  as the parabolic sub-algebra defined by the short root  $\alpha_4$ .

### 1. The Differential System and Its Infinitesimal Symmetries

Load the required DifferentialGeometry packages.

```
[M > with(DifferentialGeometry): with(LieAlgebras): with(ExteriorDifferentialSystems):
```

Create a 15-dimensional manifold. Our coordinate labels coincide with those of Cartan [2].

```
[> DGsetup([z, x1, x2, x3, x4, y1, y2, y3, y4, x12, x13, x14, x23, x24, x34], M);  
frame name: M
```

(2.1

Define a list of 7 one-forms. These are the generators of our Pfaffian system  $\Omega$ .

$$\begin{aligned} \text{M} > \text{o} &:= \text{evalDG}(\text{dz} - \text{y1}*\text{dx1} - \text{y2}*\text{dx2} - \text{y3}*\text{dx3} - \text{y4}*\text{dx4}); \\ \text{o} &:= \text{dz} - \text{y1} \text{ dx1} - \text{y2} \text{ dx2} - \text{y3} \text{ dx3} - \text{y4} \text{ dx4} \end{aligned} \quad (2.2)$$

$$\begin{aligned} \text{M} > \text{o12} &:= \text{evalDG}(\text{dx12} - \text{x1}*\text{dx2} + \text{x2}*\text{dx1} - \text{y3}*\text{dy4} + \text{y4}*\text{dy3}); \\ \text{o12} &:= \text{x2} \text{ dx1} - \text{x1} \text{ dx2} + \text{y4} \text{ dy3} - \text{y3} \text{ dy4} + \text{dx12} \end{aligned} \quad (2.3)$$

$$\begin{aligned} \text{M} > \text{o13} &:= \text{evalDG}(\text{dx13} - \text{x1}*\text{dx3} + \text{x3}*\text{dx1} + \text{y2}*\text{dy4} - \text{y4}*\text{dy2}); \\ \text{o13} &:= \text{x3} \text{ dx1} - \text{x1} \text{ dx3} - \text{y4} \text{ dy2} + \text{y2} \text{ dy4} + \text{dx13} \end{aligned} \quad (2.4)$$

$$\begin{aligned} \text{M} > \text{o14} &:= \text{evalDG}(\text{dx14} - \text{x1}*\text{dx4} + \text{x4}*\text{dx1} - \text{y2}*\text{dy3} + \text{y3}*\text{dy2}); \\ \text{o14} &:= \text{x4} \text{ dx1} - \text{x1} \text{ dx4} + \text{y3} \text{ dy2} - \text{y2} \text{ dy3} + \text{dx14} \end{aligned} \quad (2.5)$$

$$\begin{aligned} \text{M} > \text{o23} &:= \text{evalDG}(\text{dx23} - \text{x2}*\text{dx3} + \text{x3}*\text{dx2} - \text{y1}*\text{dy4} + \text{y4}*\text{dy1}); \\ \text{o23} &:= \text{x3} \text{ dx2} - \text{x2} \text{ dx3} + \text{y4} \text{ dy1} - \text{y1} \text{ dy4} + \text{dx23} \end{aligned} \quad (2.6)$$

$$\begin{aligned} \text{M} > \text{o24} &:= \text{evalDG}(\text{dx24} - \text{x2}*\text{dx4} + \text{x4}*\text{dx2} + \text{y1}*\text{dy3} - \text{y3}*\text{dy1}); \\ \text{o24} &:= \text{x4} \text{ dx2} - \text{x2} \text{ dx4} - \text{y3} \text{ dy1} + \text{y1} \text{ dy3} + \text{dx24} \end{aligned} \quad (2.7)$$

$$\begin{aligned} \text{M} > \text{o34} &:= \text{evalDG}(\text{dx34} - \text{x3}*\text{dx4} + \text{x4}*\text{dx3} - \text{y1}*\text{dy2} + \text{y2}*\text{dy1}); \\ \text{o34} &:= \text{x4} \text{ dx3} - \text{x3} \text{ dx4} + \text{y2} \text{ dy1} - \text{y1} \text{ dy2} + \text{dx34} \end{aligned} \quad (2.8)$$

The Pfaffian system is:

$$\begin{aligned} \text{M} > \text{Omega} &:= [\text{o}, \text{o12}, \text{o13}, \text{o14}, \text{o23}, \text{o24}, \text{o34}]; \\ \Omega &:= [\text{dz} - \text{y1} \text{ dx1} - \text{y2} \text{ dx2} - \text{y3} \text{ dx3} - \text{y4} \text{ dx4}, \text{x2} \text{ dx1} - \text{x1} \text{ dx2} + \text{y4} \text{ dy3} - \text{y3} \text{ dy4} + \text{dx12}, \text{x3} \text{ dx1} - \text{x1} \text{ dx3} - \text{y4} \text{ dy2} + \text{y2} \text{ dy4} + \text{dx13}, \\ &\quad \text{x4} \text{ dx1} - \text{x1} \text{ dx4} + \text{y3} \text{ dy2} - \text{y2} \text{ dy3} + \text{dx14}, \text{x3} \text{ dx2} - \text{x2} \text{ dx3} + \text{y4} \text{ dy1} - \text{y1} \text{ dy4} + \text{dx23}, \text{x4} \text{ dx2} - \text{x2} \text{ dx4} \\ &\quad - \text{y3} \text{ dy1} + \text{y1} \text{ dy3} + \text{dx24}, \text{x4} \text{ dx3} - \text{x3} \text{ dx4} + \text{y2} \text{ dy1} - \text{y1} \text{ dy2} + \text{dx34}] \end{aligned} \quad (2.9)$$

To calculate the infinitesimal symmetries of  $\Omega$ , one starts with a general vector field  $X$  on  $M$  and then calculates the 1st order partial differential equations imposed by the conditions  $\mathcal{L}_X(\omega) \subset \Omega$  for each of the generating 1-forms  $\omega \in \Omega$ . Since  $\dim M = 15$ , then a general vector field  $X$  depends upon 15 functions of 15 variables and the resulting system of 1st order PDE defining the symmetries is a very large system.

The difficulties with this example are twofold. Firstly, all the integrability conditions and a [reduced involutive form](#) (or

differential Grobner basis) must be calculated. So it is somewhat unreasonable to expect a direct call to [InfinitesimalSymmetriesOfEDS](#) to successfully find all the symmetries for very large systems. For such large systems of PDE, solutions can be found by Maple if there is a particular ordering of the independent variables which will simplify the determination of a reduced involutive form. But for the Pfaffian system  $\Omega$  the variables  $\{x1, x2, x3, x4, y1, y2, y3, y4\}$  and  $\{x12, x13, x14, x23, x24, x34\}$  are on "equal" footing and no ordering of the variables appears to be particularly advantageous.

Secondly, in many cases the calculation of infinitesimal symmetries for a differential system  $\Omega$  can be achieved by simultaneously calculating the symmetries of invariant systems associated to  $\Omega$  such as the derived system (or even the entire derived flag). But the derived system for  $\Omega$  is trivial so this is of no help.

Since the coefficients of our differential system are polynomial functions, about the best we can do is calculate symmetries with polynomial coefficients. We shall find a 52-dimensional Lie algebra of polynomial symmetries of total degree  $\leq 3$ . Later on we prove that the symmetry algebra of our system has dimension  $\leq 52$ , and therefore our polynomial ansatz for the symmetries has produced all symmetries.

```
[M > Gamma := InfinitesimalSymmetriesOfEDS(Omega, ansatz = "polynomial", degree = 3, output
= "list");
```

Here is the number of symmetries.

```
[alg > nops(Gamma);
```

$$52 \tag{2.10}$$

Here are just a few of the symmetries. We shall look at more of them once we have calculated the Chevalley basis.

```
[alg1 > Gamma[1];
```

$$\partial_z \tag{2.11}$$

```
[alg1 > Gamma[2];
```

$$x3 \partial_{x3} - x4 \partial_{x4} - y3 \partial_{y3} + y4 \partial_{y4} + x13 \partial_{x13} - x14 \partial_{x14} + x23 \partial_{x23} - x24 \partial_{x24} \tag{2.12}$$

```
[alg1 > Gamma[3];
```

$$\partial_{x4} - x1 \partial_{x14} - x2 \partial_{x24} - x3 \partial_{x34} \tag{2.13}$$

```
[alg1 > Gamma[4];
```

$$\tag{2.14}$$

$$x1 \partial_{x4} - y4 \partial_{y1} - x12 \partial_{x24} - x13 \partial_{x34} \quad (2.14)$$

When we calculate polynomial symmetries to a given degree, there is no guarantee that the result will define a Lie algebra of vector fields. We use the command [LieAlgebraData](#) to calculate the structure equations for our symmetries and we find that our polynomial symmetries do define a Lie algebra. Again this is a very lengthy calculation. The number of structure equations  $52 \cdot 51 / 2 = 1326$  and for each structure equation a very large system of linear equations must be solved to determine the structure constants. The following calling sequence seems to give the result about as quickly as currently possible.

For this particular calling sequence, we need a list of points on the manifold M.

```
M > pt1 := [z = 0, x1 = 0, x2 = 0, x3 = 0, x4 = 0, y1 = 0, y2 = 0, y3 = 0, y4 = 0, x12 = 0, x13 = 0, x14 = 0, x23 = 0, x24 = 0, x34 = 0];
pt1 := [z = 0, x1 = 0, x2 = 0, x3 = 0, x4 = 0, y1 = 0, y2 = 0, y3 = 0, y4 = 0, x12 = 0, x13 = 0, x14 = 0, x23 = 0, x24 = 0, x34 = 0] (2.15)
```

```
M > pt2 := [z = 1, x1 = 0, x2 = 0, x3 = 0, x4 = 0, y1 = 0, y2 = 0, y3 = 0, y4 = 0, x12 = 0, x13 = 0, x14 = 0, x23 = 0, x24 = 0, x34 = 0];
pt2 := [z = 1, x1 = 0, x2 = 0, x3 = 0, x4 = 0, y1 = 0, y2 = 0, y3 = 0, y4 = 0, x12 = 0, x13 = 0, x14 = 0, x23 = 0, x24 = 0, x34 = 0] (2.16)
```

```
M > pt3 := [z = 0, x1 = 1, x2 = 0, x3 = 0, x4 = 0, y1 = 0, y2 = 0, y3 = 0, y4 = 0, x12 = 0, x13 = 0, x14 = 0, x23 = 0, x24 = 0, x34 = 0];
pt3 := [z = 0, x1 = 1, x2 = 0, x3 = 0, x4 = 0, y1 = 0, y2 = 0, y3 = 0, y4 = 0, x12 = 0, x13 = 0, x14 = 0, x23 = 0, x24 = 0, x34 = 0] (2.17)
```

```
M > pt4 := [z = 0, x1 = 0, x2 = 1, x3 = 0, x4 = 0, y1 = 0, y2 = 0, y3 = 0, y4 = 0, x12 = 0, x13 = 0, x14 = 0, x23 = 0, x24 = 0, x34 = 0];
pt4 := [z = 0, x1 = 0, x2 = 1, x3 = 0, x4 = 0, y1 = 0, y2 = 0, y3 = 0, y4 = 0, x12 = 0, x13 = 0, x14 = 0, x23 = 0, x24 = 0, x34 = 0] (2.18)
```

The keyword argument `calculate = "onebyone"` means that the structure equations will be calculated one at a time. The default keyword value `calculate = "allatonce"` generally works faster for Lie algebras of modest size.

```
alg > LD1 := LieAlgebraData(Gamma, alg1, initialpointlist = [pt1, pt2, pt3, pt4],
    calculate = "onebyone");
alg > DGsetup(LD1);
Lie algebra: alg1 (2.19)
```

From the Cartan criterion for semisimplicity, we easily check that this 52-dimensional Lie algebra is semi-simple.

```
alg1 > Query("Semisimple"); (2.20)
```

*true*

(2.20)

## 2. Algebraic Classification of the Infinitesimal Symmetries and the Chevalley Basis

Now we use the command `ClassifyComplexSemiSimpleLieAlgebra` to classify the symmetry algebra of the Cartan differential system. We use `infolevel` to track the progress of this calculation.

```
[alg > infolevel[ClassifyComplexSemiSimpleLieAlgebra] := 1;
                               infolevel DifferentialGeometry:-LieAlgebras:-ClassifyComplexSemiSimpleLieAlgebra := 1] (3.1)
```

```
[alg1 > ClassifyComplexSemiSimpleLieAlgebra(alg1, properties = 'P');
Step 1. Find a Cartan subalgebra
Step 2. Find the root space decomposition
Step 3. Find the positive and simple roots
Step 4. Find the Cartan matrix
Step 5. Put the Cartan matrix in standard form
                               "F4"] (3.2)
```

With the keyword argument `properties = P`, the variable `P` is assigned as a [record](#) containing all the intermediate results used in the classification. These results are stored as exports of the record `P`.

```
[M > exports(P);
                               CartanSubalgebra, SimpleRoots, RootSpaceDecomposition, PositiveRoots] (3.3)
```

Here are the simple roots.

```
[alg1 > P:-SimpleRoots;
                               
$$\left[ \begin{bmatrix} 1 \\ -2 \\ 0 \\ 1 \end{bmatrix}, \begin{bmatrix} 0 \\ 1 \\ -1 \\ -2 \end{bmatrix}, \begin{bmatrix} 0 \\ 0 \\ 0 \\ 1 \end{bmatrix}, \begin{bmatrix} 0 \\ 0 \\ 1 \\ 0 \end{bmatrix} \right] ] (3.4)$$

```

Since the simple roots are all real, we can conclude that the symmetry algebra  $\Gamma$  of the Cartan Pfaffian system is the split

### real form of the exceptional Lie algebra $f_4$

For the split real form of a simple Lie algebra there is a preferred basis called the [Chevalley basis](#). We calculate the Chevalley basis for the abstract Lie algebra **alg**.

$$\begin{aligned} \text{alg1} > \text{ChevBasis} &:= \text{ChevalleyBasis}(\mathbf{P}); \\ \text{ChevBasis} &:= [ -e_{10}, -e_{45}, e_{10} - e_{21} + 2e_{45}, e_2 + 2e_{21} - e_{45}, -e_4, -e_{39}, e_{22}, e_{49}, e_{12}, e_{41}, -e_7, -e_3, e_{20}, e_{43}, \\ &\quad -e_5, -e_9, e_{46}, e_{47}, -e_{15}, e_{38}, -e_{25}, -e_{17}, e_{26}, -e_{27}, e_{28}, -e_{24}, -e_{42}, -e_6, -e_{31}, -e_{16}, -e_{34}, -e_{18}, \\ &\quad -e_{36}, -2e_{14}, -e_1, -e_{35}, -e_{11}, -e_{19}, -e_{32}, -e_{44}, -e_8, -e_{37}, -e_{29}, -e_{23}, -e_{50}, -e_{40}, -e_{30}, \\ &\quad -e_{52}, -e_{51}, -e_{48}, -e_{13}, -e_{33} ] \end{aligned} \quad (3.5)$$

We now rewrite the Chevalley basis in terms of infinitesimal generators for the symmetry algebra. The command **convert** (**ChevBasis**, **DGlist**) returns a list of lists, representing the components of the Chevalley basis.

$$\text{alg2} > \text{Gamma1} := \text{DGzip}(\text{convert}(\text{ChevBasis}, \text{DGlist}), \text{Gamma});$$

We now check some of the defining properties of the Chevalley basis in terms of infinitesimal symmetry generators.

(1) The first 4 elements in the Chevalley basis span the Cartan subalgebra.

$$\begin{aligned} \text{alg2} > \text{Gamma1}[1]; \\ x_1 \partial_{x_1} - x_4 \partial_{x_4} - y_1 \partial_{y_1} + y_4 \partial_{y_4} + x_{12} \partial_{x_{12}} + x_{13} \partial_{x_{13}} - x_{24} \partial_{x_{24}} - x_{34} \partial_{x_{34}} \end{aligned} \quad (3.6)$$

$$\begin{aligned} \text{alg2} > \text{Gamma1}[2]; \\ -z \partial_z - x_1 \partial_{x_1} - x_2 \partial_{x_2} - y_3 \partial_{y_3} - y_4 \partial_{y_4} - 2x_{12} \partial_{x_{12}} - x_{13} \partial_{x_{13}} - x_{14} \partial_{x_{14}} - x_{23} \partial_{x_{23}} - x_{24} \partial_{x_{24}} \end{aligned} \quad (3.7)$$

$$\begin{aligned} \text{alg2} > \text{Gamma1}[3]; \\ z \partial_z + 2x_2 \partial_{x_2} + y_1 \partial_{y_1} - y_2 \partial_{y_2} + y_3 \partial_{y_3} + y_4 \partial_{y_4} + 2x_{12} \partial_{x_{12}} + 2x_{23} \partial_{x_{23}} + 2x_{24} \partial_{x_{24}} \end{aligned} \quad (3.8)$$

$$\begin{aligned} \text{alg2} > \text{Gamma1}[4]; \\ z \partial_z + x_1 \partial_{x_1} - x_2 \partial_{x_2} + x_3 \partial_{x_3} + x_4 \partial_{x_4} + 2y_2 \partial_{y_2} + 2x_{13} \partial_{x_{13}} + 2x_{14} \partial_{x_{14}} + 2x_{34} \partial_{x_{34}} \end{aligned} \quad (3.9)$$

These vector fields define an abelian subalgebra:

$$\begin{aligned} \text{alg1} > \text{LieAlgebraData}(\text{Gamma1}[1..4]); \\ [e_1, e_2] = 0, [e_1, e_3] = 0, [e_1, e_4] = 0, [e_2, e_3] = 0, [e_2, e_4] = 0, [e_3, e_4] = 0 \end{aligned} \quad (3.10)$$

(2) The remaining vectors all define the root spaces. For example, the vector **Gamma1[33]** determines the root space for the root  $[-1, -1, 2, 0]$ .

$$\begin{aligned}
 & \mathbf{M} > \mathbf{LB} := \mathbf{LieDerivative}(\mathbf{Gamma1}[1..4], \mathbf{Gamma1}[33]); \\
 & \mathbf{LB} := \left[ -x_{24} z \partial_z + (x_{12} x_4 - x_{14} x_2 + y_3 z) \partial_{x_1} - x_2 x_{24} \partial_{x_2} - (x_{34} x_2 + x_4 x_{23} + y_1 z) \partial_{x_3} - x_{24} x_4 \partial_{x_4} - x_{24} y_1 \partial_{y_1} - \right. \\
 & \quad \left( x_1 x_4 y_1 + x_2 x_4 y_2 + x_3 x_4 y_3 + y_4 x_4^2 - x_{14} y_1 - x_{34} y_3 - x_4 z \right) \partial_{y_2} - x_{24} y_3 \partial_{y_3} + (x_1 x_2 y_1 + y_2 x_2^2 + x_2 x_3 y_3 \\
 & \quad + x_4 x_2 y_4 - x_{12} y_1 - x_2 z + x_{23} y_3) \partial_{y_4} - x_{12} x_{24} \partial_{x_{12}} - (x_1 y_1 z + x_2 y_2 z + y_3 x_3 z + x_4 y_4 z + x_{12} x_{34} + x_{14} x_{23} \\
 & \quad - z^2) \partial_{x_{13}} - x_{14} x_{24} \partial_{x_{14}} - x_{23} x_{24} \partial_{x_{23}} - x_{24}^2 \partial_{x_{24}} - x_{24} x_{34} \partial_{x_{34}}, -x_{24} z \partial_z + (x_{12} x_4 - x_{14} x_2 + y_3 z) \partial_{x_1} \\
 & \quad - x_2 x_{24} \partial_{x_2} - (x_{34} x_2 + x_4 x_{23} + y_1 z) \partial_{x_3} - x_{24} x_4 \partial_{x_4} - x_{24} y_1 \partial_{y_1} - (x_1 x_4 y_1 + x_2 x_4 y_2 + x_3 x_4 y_3 + y_4 x_4^2 \\
 & \quad - x_{14} y_1 - x_{34} y_3 - x_4 z) \partial_{y_2} - x_{24} y_3 \partial_{y_3} + (x_1 x_2 y_1 + y_2 x_2^2 + x_2 x_3 y_3 + x_4 x_2 y_4 - x_{12} y_1 - x_2 z + x_{23} y_3) \partial_{y_4} \\
 & \quad - x_{12} x_{24} \partial_{x_{12}} - (x_1 y_1 z + x_2 y_2 z + y_3 x_3 z + x_4 y_4 z + x_{12} x_{34} + x_{14} x_{23} - z^2) \partial_{x_{13}} - x_{14} x_{24} \partial_{x_{14}} - x_{23} x_{24} \partial_{x_{23}} \\
 & \quad - x_{24}^2 \partial_{x_{24}} - x_{24} x_{34} \partial_{x_{34}}, 2 x_{24} z \partial_z - (2 x_{12} x_4 - 2 x_{14} x_2 + 2 y_3 z) \partial_{x_1} + 2 x_2 x_{24} \partial_{x_2} + (2 x_{34} x_2 + 2 x_4 x_{23} + 2 y_1 z \\
 & \quad ) \partial_{x_3} + 2 x_{24} x_4 \partial_{x_4} + 2 x_{24} y_1 \partial_{y_1} + (2 x_1 x_4 y_1 + 2 x_2 x_4 y_2 + 2 x_3 x_4 y_3 + 2 y_4 x_4^2 - 2 x_{14} y_1 - 2 x_{34} y_3 - 2 x_4 z) \partial_{y_2} \\
 & \quad + 2 x_{24} y_3 \partial_{y_3} - (2 x_1 x_2 y_1 + 2 y_2 x_2^2 + 2 x_2 x_3 y_3 + 2 x_4 x_2 y_4 - 2 x_{12} y_1 - 2 x_2 z + 2 x_{23} y_3) \partial_{y_4} + 2 x_{12} x_{24} \partial_{x_{12}} + \\
 & \quad \left( 2 x_1 y_1 z + 2 x_2 y_2 z + 2 y_3 x_3 z + 2 x_4 y_4 z + 2 x_{12} x_{34} + 2 x_{14} x_{23} - 2 z^2 \right) \partial_{x_{13}} + 2 x_{14} x_{24} \partial_{x_{14}} + 2 x_{23} x_{24} \partial_{x_{23}} \\
 & \quad \left. + 2 x_{24}^2 \partial_{x_{24}} + 2 x_{24} x_{34} \partial_{x_{34}}, 0 \partial_z \right]
 \end{aligned} \tag{3.11}$$

$$\begin{aligned}
 & \mathbf{alg2} > \mathbf{GetComponents}(\mathbf{LB}, \mathbf{Gamma1}[33..33]); \\
 & \quad \quad \quad [[-1], [-1], [2], [0]]
 \end{aligned} \tag{3.12}$$

(3) The first 4 vectors in the Chevalley basis after the Cartan subalgebra give the root spaces for the simple roots.

$$\begin{aligned}
 & \mathbf{alg2} > \mathbf{Gamma1}[5]; \\
 & \quad \quad \quad -x_1 \partial_{x_4} + y_4 \partial_{y_1} + x_{12} \partial_{x_{24}} + x_{13} \partial_{x_{34}}
 \end{aligned} \tag{3.13}$$

$$\begin{aligned}
 & \mathbf{alg2} > \mathbf{Gamma1}[6]; \\
 & \quad \quad \quad -\partial_{x_{12}}
 \end{aligned} \tag{3.14}$$

**alg2 > Gamma1[7];**

$$\begin{aligned}
 & -x_2 z \partial_z - (x_1 x_2 + x_{12}) \partial_{x_1} - x_2^2 \partial_{x_2} - (x_2 x_3 - x_{23}) \partial_{x_3} - (x_2 x_4 - x_{24}) \partial_{x_4} - x_2 y_1 \partial_{y_1} + (2 x_1 y_1 + x_2 y_2 + 2 x_3 y_3 \\
 & + 2 x_4 y_4 - z) \partial_{y_2} - x_2 y_3 \partial_{y_3} - x_2 y_4 \partial_{y_4} - x_{12} x_2 \partial_{x_{12}} - (x_1 x_{23} + x_{12} x_3 - y_4 z) \partial_{x_{13}} - (x_1 x_{24} + x_{12} x_4 + y_3 z) \partial_{x_{14}} \\
 & - x_2 x_{23} \partial_{x_{23}} - x_2 x_{24} \partial_{x_{24}} + (x_4 x_{23} - x_{24} x_3 + y_1 z) \partial_{x_{34}}
 \end{aligned} \tag{3.15}$$

**M > Gamma1[8];**

$$\begin{aligned}
 & - (x_1 x_{34} + x_{13} x_4 - x_{14} x_3) \partial_z - x_1 y_2 \partial_{x_1} + (x_1 y_1 + x_3 y_3 + x_4 y_4 + z) \partial_{x_2} - x_3 y_2 \partial_{x_3} - x_4 y_2 \partial_{x_4} - (y_1 y_2 + x_{34}) \partial_{y_1} \\
 & - y_2^2 \partial_{y_2} + (-y_2 y_3 + x_{14}) \partial_{y_3} - (y_2 y_4 + x_{13}) \partial_{y_4} + (x_1^2 y_1 + x_1 x_2 y_2 + x_1 x_3 y_3 + x_1 x_4 y_4 - x_1 z + x_{13} y_3 + x_{14} y_4 \\
 & ) \partial_{x_{12}} - x_{13} y_2 \partial_{x_{13}} - x_{14} y_2 \partial_{x_{14}} - (x_1 x_3 y_1 + x_2 x_3 y_2 + x_3^2 y_3 + x_3 x_4 y_4 - x_{13} y_1 - x_3 z + x_{34} y_4) \partial_{x_{23}} - (x_1 x_4 y_1 \\
 & + x_2 x_4 y_2 + x_3 x_4 y_3 + y_4 x_4^2 - x_{14} y_1 - x_{34} y_3 - x_4 z) \partial_{x_{24}} - x_{34} y_2 \partial_{x_{34}}
 \end{aligned} \tag{3.16}$$

From the Lie brackets of the Cartan subalgebra and these 4 vectors, one reads off the Cartan matrix:

**M > Matrix(4, 4, (i, j) -> GetComponents(LieBracket(Gamma1[i], Gamma1[j + 4]), Gamma1[j + 4 .. j + 4])[1]);**

$$\begin{bmatrix} 2 & -1 & 0 & 0 \\ -1 & 2 & -1 & 0 \\ 0 & -2 & 2 & -1 \\ 0 & 0 & -1 & 2 \end{bmatrix}$$

(3.17)

(4) The positive root spaces are defined by the vectors Gamma1[5 .. 28] and the negative root spaces by the vectors Gamma1[29 .. 52].

The first element of the basis for the Cartan subalgebra, the vector associated to the 1st simple root, and the vector associated to the 1st negative simple root together generate the algebra  $sl(2)$ .

**M > Gamma1[29];**

$$x_4 \partial_{x_1} - y_1 \partial_{y_4} - x_{24} \partial_{x_{12}} - x_{34} \partial_{x_{13}} \tag{3.18}$$

**M > Gamma1[1];**

(3.19)



$$x1 \partial_{x1} - x4 \partial_{x4} - y1 \partial_{y1} + y4 \partial_{y4} + x12 \partial_{x12} + x13 \partial_{x13} - x24 \partial_{x24} - x34 \partial_{x34} \quad (3.19)$$

**M > Gamma1[5];**

$$-x1 \partial_{x4} + y4 \partial_{y1} + x12 \partial_{x24} + x13 \partial_{x34} \quad (3.20)$$

**M > LieAlgebraData([Gamma1[29], Gamma1[1], Gamma1[5]]);**

$$[e1, e2] = 2 e1, [e1, e3] = e2, [e2, e3] = 2 e3 \quad (3.21)$$

The same is true for the other basis elements of the Cartan subalgebra:

**M > Gamma1[30];**

$$\begin{aligned} & -x12 z \partial_z - x1 x12 \partial_{x1} - x12 x2 \partial_{x2} + (x1 x23 - x13 x2 - y4 z) \partial_{x3} + (x1 x24 - x14 x2 + y3 z) \partial_{x4} - (x1 x2 y1 + y2 x2^2 \\ & + x2 x3 y3 + x4 x2 y4 - x2 z + x23 y3 + x24 y4) \partial_{y1} + (x1^2 y1 + x1 x2 y2 + x1 x3 y3 + x1 x4 y4 - x1 z + x13 y3 + x14 y4 \\ & ) \partial_{y2} - x12 y3 \partial_{y3} - x12 y4 \partial_{y4} - x12^2 \partial_{x12} - x12 x13 \partial_{x13} - x12 x14 \partial_{x14} - x12 x23 \partial_{x23} - x12 x24 \partial_{x24} + (x1 y1 z \\ & + x2 y2 z + y3 x3 z + x4 y4 z - x13 x24 + x14 x23 - z^2) \partial_{x34} \end{aligned} \quad (3.22)$$

**M > Gamma1[2];**

$$-z \partial_z - x1 \partial_{x1} - x2 \partial_{x2} - y3 \partial_{y3} - y4 \partial_{y4} - 2 x12 \partial_{x12} - x13 \partial_{x13} - x14 \partial_{x14} - x23 \partial_{x23} - x24 \partial_{x24} \quad (3.23)$$

**M > Gamma1[6];**

$$-\partial_{x12} \quad (3.24)$$

**M > LieAlgebraData([Gamma1[30], Gamma1[2], Gamma1[6]]);**

$$[e1, e2] = 2 e1, [e1, e3] = e2, [e2, e3] = 2 e3 \quad (3.25)$$

**M > LieAlgebraData([Gamma1[31], Gamma1[3], Gamma1[7]]);**

$$[e1, e2] = 2 e1, [e1, e3] = e2, [e2, e3] = 2 e3 \quad (3.26)$$

**M > LieAlgebraData([Gamma1[32], Gamma1[4], Gamma1[8]]);**

$$[e1, e2] = 2 e1, [e1, e3] = e2, [e2, e3] = 2 e3 \quad (3.27)$$

Other properties of the [Chevalley basis](#) are similarly checked.

### 3. Symbol Algebra and Tanaka Prolongation

To any differential system  $\Omega$  one can associate a graded nilpotent Lie algebra, called the [symbol algebra](#)  $\mathfrak{m}(\Omega)$  of  $\Omega$ .

From the succession of graded derivations of  $\mathfrak{m}(\Omega)$ , Tanaka defined the [prolongation](#)  $pr(\mathfrak{m}(\Omega))$  of  $\mathfrak{m}(\Omega)$  and proved

$$\dim(\text{sym}(\Omega)) \leq \dim(pr(\mathfrak{m}(\Omega)))$$

See Yamaguchi [3] for an excellent review of the symbol algebra and its Tanaka prolongation. In this section we calculate the symbol algebra of the Cartan differential system analyzed above. The Tanaka prolongation is seen to be a 52 dimensional Lie algebra. *This proves that the symmetry algebra of polynomial vector fields which we computed in Section 1 is actually the full symmetry algebra of the Cartan differential system.*

To calculate the [symbol algebra](#), we need the vector field system  $\Delta$  dual to  $\Omega$ , defined as the distribution which [annihilates](#) the generating 1-forms for the Pfaffian system. We compute a basis  $\Delta 0$  for  $\Delta$ .

**M > Delta0 := Annihilator(Omega);**

$$\Delta 0 := \left[ \begin{aligned} & -\frac{y^2}{x_1 y_1 + x_2 y_2 + x_3 y_3 + x_4 y_4} \partial_{x_1} + \frac{y_1}{x_1 y_1 + x_2 y_2 + x_3 y_3 + x_4 y_4} \partial_{x_2} \\ & -\frac{x_3 y_1}{y_4 (x_1 y_1 + x_2 y_2 + x_3 y_3 + x_4 y_4)} \partial_{y_1} - \frac{x_3 y_2}{y_4 (x_1 y_1 + x_2 y_2 + x_3 y_3 + x_4 y_4)} \partial_{y_2} \\ & -\frac{x_3 y_3 + x_4 y_4}{y_4 (x_1 y_1 + x_2 y_2 + x_3 y_3 + x_4 y_4)} \partial_{y_3} + \partial_{x_{12}}, \frac{y_1}{y_4} \partial_{y_1} + \frac{y_2}{y_4} \partial_{y_2} + \frac{y_3}{y_4} \partial_{y_3} + \partial_{y_4}, \\ & -\frac{y^3}{x_1 y_1 + x_2 y_2 + x_3 y_3 + x_4 y_4} \partial_{x_1} + \frac{y_1}{x_1 y_1 + x_2 y_2 + x_3 y_3 + x_4 y_4} \partial_{x_3} + \frac{x_2 y_1}{y_4 (x_1 y_1 + x_2 y_2 + x_3 y_3 + x_4 y_4)} \partial_{y_1} \\ & + \frac{x_2 y_2 + x_4 y_4}{y_4 (x_1 y_1 + x_2 y_2 + x_3 y_3 + x_4 y_4)} \partial_{y_2} + \frac{x_2 y_3}{y_4 (x_1 y_1 + x_2 y_2 + x_3 y_3 + x_4 y_4)} \partial_{y_3} + \partial_{x_{13}}, \\ & -\frac{y^3}{x_1 y_1 + x_2 y_2 + x_3 y_3 + x_4 y_4} \partial_{x_2} + \frac{y^2}{x_1 y_1 + x_2 y_2 + x_3 y_3 + x_4 y_4} \partial_{x_3} - \frac{x_1 y_1 + x_4 y_4}{y_4 (x_1 y_1 + x_2 y_2 + x_3 y_3 + x_4 y_4)} \partial_{y_1} \\ & -\frac{x_1 y_2}{y_4 (x_1 y_1 + x_2 y_2 + x_3 y_3 + x_4 y_4)} \partial_{y_2} - \frac{x_1 y_3}{y_4 (x_1 y_1 + x_2 y_2 + x_3 y_3 + x_4 y_4)} \partial_{y_3} + \partial_{x_{23}}, -\frac{y_1}{x_4} \partial_z \\ & -\frac{x_1 y_1 + x_4 y_4}{(x_1 y_1 + x_2 y_2 + x_3 y_3 + x_4 y_4) x_4} \partial_{x_1} - \frac{x_2 y_1}{(x_1 y_1 + x_2 y_2 + x_3 y_3 + x_4 y_4) x_4} \partial_{x_2} \end{aligned} \right] \quad (4.1)$$

$$\begin{aligned}
& - \frac{x_3 y_1}{(x_1 y_1 + x_2 y_2 + x_3 y_3 + x_4 y_4) x_4} \partial_{x_3} - \frac{x_3}{x_1 y_1 + x_2 y_2 + x_3 y_3 + x_4 y_4} \partial_{y_2} + \frac{x_2}{x_1 y_1 + x_2 y_2 + x_3 y_3 + x_4 y_4} \partial_{y_3} \\
& + \partial_{x_{14}}, - \frac{y_2}{x_4} \partial_z - \frac{x_1 y_2}{(x_1 y_1 + x_2 y_2 + x_3 y_3 + x_4 y_4) x_4} \partial_{x_1} - \frac{x_2 y_2 + x_4 y_4}{(x_1 y_1 + x_2 y_2 + x_3 y_3 + x_4 y_4) x_4} \partial_{x_2} \\
& - \frac{x_3 y_2}{(x_1 y_1 + x_2 y_2 + x_3 y_3 + x_4 y_4) x_4} \partial_{x_3} + \frac{x_3}{x_1 y_1 + x_2 y_2 + x_3 y_3 + x_4 y_4} \partial_{y_1} - \frac{x_1}{x_1 y_1 + x_2 y_2 + x_3 y_3 + x_4 y_4} \partial_{y_3} \\
& + \partial_{x_{24}}, - \frac{y_3}{x_4} \partial_z - \frac{x_1 y_3}{(x_1 y_1 + x_2 y_2 + x_3 y_3 + x_4 y_4) x_4} \partial_{x_1} - \frac{x_2 y_3}{(x_1 y_1 + x_2 y_2 + x_3 y_3 + x_4 y_4) x_4} \partial_{x_2} \\
& - \frac{x_3 y_3 + x_4 y_4}{(x_1 y_1 + x_2 y_2 + x_3 y_3 + x_4 y_4) x_4} \partial_{x_3} - \frac{x_2}{x_1 y_1 + x_2 y_2 + x_3 y_3 + x_4 y_4} \partial_{y_1} + \frac{x_1}{x_1 y_1 + x_2 y_2 + x_3 y_3 + x_4 y_4} \partial_{y_2} \\
& + \partial_{x_{34}}, \frac{x_1 y_1 + x_2 y_2 + x_3 y_3 + x_4 y_4}{x_4} \partial_z + \frac{x_1}{x_4} \partial_{x_1} + \frac{x_2}{x_4} \partial_{x_2} + \frac{x_3}{x_4} \partial_{x_3} + \partial_{x_4} \Big]
\end{aligned}$$

We put this result into canonical form with respect to a judiciously chosen ordering of the basis vectors for the tangent space.

$$\begin{aligned}
& \mathbf{M} > \text{Delta1} := \text{CanonicalBasis}(\text{Delta0}, [\text{D\_y1}, \text{D\_y2}, \text{D\_y3}, \text{D\_y4}, \text{D\_x1}, \text{D\_x2}, \text{D\_x3}, \text{D\_x4}, \\
& \quad \text{D\_x12}, \text{D\_x13}, \text{D\_x14}, \text{D\_x23}, \text{D\_x24}, \text{D\_x34}, \text{D\_z}]); \\
& \Delta I := \left[ \partial_{y_1} - y_4 \partial_{x_{23}} + y_3 \partial_{x_{24}} - y_2 \partial_{x_{34}}, \partial_{y_2} + y_4 \partial_{x_{13}} - y_3 \partial_{x_{14}} + y_1 \partial_{x_{34}}, \partial_{y_3} - y_4 \partial_{x_{12}} + y_2 \partial_{x_{14}} - y_1 \partial_{x_{24}}, \partial_{y_4} + y_3 \partial_{x_{12}} \right. \\
& \quad - y_2 \partial_{x_{13}} + y_1 \partial_{x_{23}}, y_1 \partial_z + \partial_{x_1} - x_2 \partial_{x_{12}} - x_3 \partial_{x_{13}} - x_4 \partial_{x_{14}}, y_2 \partial_z + \partial_{x_2} + x_1 \partial_{x_{12}} - x_3 \partial_{x_{23}} - x_4 \partial_{x_{24}}, y_3 \partial_z + \partial_{x_3} \\
& \quad \left. + x_1 \partial_{x_{13}} + x_2 \partial_{x_{23}} - x_4 \partial_{x_{34}}, y_4 \partial_z + \partial_{x_4} + x_1 \partial_{x_{14}} + x_2 \partial_{x_{24}} + x_3 \partial_{x_{34}} \right]
\end{aligned} \tag{4.2}$$

The symbol algebra is defined pointwise so we need to specify a point at which the symbol algebra will be calculated. (A distribution is called *strongly regular* if the symbol algebras at different points are all isomorphic as graded nilpotent Lie algebras.)

$$\begin{aligned}
& \mathbf{M} > \text{pt1} := [\text{z} = 0, \text{x1} = 0, \text{x2} = 0, \text{x3} = 0, \text{x4} = 0, \text{y1} = 0, \text{y2} = 0, \text{y3} = 0, \text{y4} = 0, \text{x12} = \\
& \quad 0, \text{x13} = 0, \text{x14} = 0, \text{x23} = 0, \text{x24} = 0, \text{x34} = 0]; \\
& \text{pt1} := [\text{z} = 0, \text{x1} = 0, \text{x2} = 0, \text{x3} = 0, \text{x4} = 0, \text{y1} = 0, \text{y2} = 0, \text{y3} = 0, \text{y4} = 0, \text{x12} = 0, \text{x13} = 0, \text{x14} = 0, \text{x23} = 0, \text{x24} = 0, \text{x34} = 0]
\end{aligned} \tag{4.3}$$

SymbolAlgebra calculates a basis for the symbol algebra. With the keyword argument **order = "Increasing"**, the basis is given in increasing order by weight, starting from the most negative weight and ending with the vectors of weight -1.

```

M > LD3 := SymbolAlgebra(Delta1, pt1, symalg, order = "Increasing");
LD3 := [e1, e2] = 0, [e1, e3] = 0, [e1, e4] = 0, [e1, e5] = 0, [e1, e6] = 0, [e1, e7] = 0, [e1, e8] = 0, [e1, e9] = 0, [e1, e10] = 0,
[e1, e11] = 0, [e1, e12] = 0, [e1, e13] = 0, [e1, e14] = 0, [e1, e15] = 0, [e2, e3] = 0, [e2, e4] = 0, [e2, e5] = 0, [e2, e6] = 0,
[e2, e7] = 0, [e2, e8] = 0, [e2, e9] = 0, [e2, e10] = 0, [e2, e11] = 0, [e2, e12] = 0, [e2, e13] = 0, [e2, e14] = 0, [e2, e15]
] = 0, [e3, e4] = 0, [e3, e5] = 0, [e3, e6] = 0, [e3, e7] = 0, [e3, e8] = 0, [e3, e9] = 0, [e3, e10] = 0, [e3, e11] = 0, [e3, e12
] = 0, [e3, e13] = 0, [e3, e14] = 0, [e3, e15] = 0, [e4, e5] = 0, [e4, e6] = 0, [e4, e7] = 0, [e4, e8] = 0, [e4, e9] = 0, [e4, e10
] = 0, [e4, e11] = 0, [e4, e12] = 0, [e4, e13] = 0, [e4, e14] = 0, [e4, e15] = 0, [e5, e6] = 0, [e5, e7] = 0, [e5, e8] = 0, [e5, e9
] = 0, [e5, e10] = 0, [e5, e11] = 0, [e5, e12] = 0, [e5, e13] = 0, [e5, e14] = 0, [e5, e15] = 0, [e6, e7] = 0, [e6, e8] = 0, [e6,
e9] = 0, [e6, e10] = 0, [e6, e11] = 0, [e6, e12] = 0, [e6, e13] = 0, [e6, e14] = 0, [e6, e15] = 0, [e7, e8] = 0, [e7, e9] = 0,
[e7, e10] = 0, [e7, e11] = 0, [e7, e12] = 0, [e7, e13] = 0, [e7, e14] = 0, [e7, e15] = 0, [e8, e9] = e1, [e8, e10] = e2, [e8, e11
] = e3, [e8, e12] = e4, [e8, e13] = 0, [e8, e14] = 0, [e8, e15] = 0, [e9, e10] = e5, [e9, e11] = e6, [e9, e12] = 0, [e9, e13
] = e4, [e9, e14] = 0, [e9, e15] = 0, [e10, e11] = e7, [e10, e12] = 0, [e10, e13] = 0, [e10, e14] = e4, [e10, e15] = 0, [e11,
e12] = 0, [e11, e13] = 0, [e11, e14] = 0, [e11, e15] = e4, [e12, e13] = e7, [e12, e14] = -e6, [e12, e15] = e5, [e13, e14
] = e3, [e13, e15] = -e2, [e14, e15] = e1

```

Initialize and print the table of Lie brackets.

```

M > DGsetup(LD3, [X], [O]);
Lie algebra: symalg
symalg > MultiplicationTable("LieTable");

```

symalg	$X1$	$X2$	$X3$	$X4$	$X5$	$X6$	$X7$	$X8$	$X9$	$X10$	$X11$	$X12$	$X13$	$X14$	$X15$
$X1$	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
$X2$	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
$X3$	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
$X4$	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
$X5$	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
$X6$	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
$X7$	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
$X8$	0	0	0	0	0	0	0	0	$X1$	$X2$	$X3$	$X4$	0	0	0
$X9$	0	0	0	0	0	0	0	$-X1$	0	$X5$	$X6$	0	$X4$	0	0
$X10$	0	0	0	0	0	0	0	$-X2$	$-X5$	0	$X7$	0	0	$X4$	0
$X11$	0	0	0	0	0	0	0	$-X3$	$-X6$	$-X7$	0	0	0	0	$X4$
$X12$	0	0	0	0	0	0	0	$-X4$	0	0	0	0	$X7$	$-X6$	$X5$
$X13$	0	0	0	0	0	0	0	0	$-X4$	0	0	$-X7$	0	$X3$	$-X2$
$X14$	0	0	0	0	0	0	0	0	0	$-X4$	0	$X6$	$-X3$	0	$X1$
$X15$	0	0	0	0	0	0	0	0	0	0	$-X4$	$-X5$	$X2$	$-X1$	0

(4.6)

Here are the grading weights for each basis element.

```
[symalg > Tools:-DGinfo("Grading");
      [-2, -2, -2, -2, -2, -2, -2, -2, -1, -1, -1, -1, -1, -1, -1, -1]]
```

(4.7)

This means that the symbol algebra is  $\mathfrak{m}(\Delta) = \mathfrak{g}_{-2} \oplus \mathfrak{g}_{-1}$ , where  $\mathfrak{g}_{-2} = \text{span}\{X1, X2, X3, X4, X5, X6, X7\}$  and  $\mathfrak{g}_{-1} = \text{span}\{X8, X9, X10, X11, X12, X13, X14\}$ .

The [prolongation](#) of  $\mathfrak{m}(\Delta)$  is, roughly speaking, a sequence of vector spaces determined by the graded derivations of

$m(\Delta)$ .

```
[symalg > infolevel[TanakaProlongation] := 2;
```

```
infolevelDifferentialGeometry:-LieAlgebras:-TanakaProlongation := 2
```

(4.8)

The Tanaka prolongation of our symbol algbrea may take a several minutes to compute (it will terminate at order 2).

```
[symalg > TanPr := TanakaProlongation(symalg, 4, prsym):
```

```
m:
```

```
[[X1, X2, X3, X4, X5, X6, X7], [X8, X9, X10, X11, X12, X13, X14, X15]]
[-2, -1]
```

The Tanaka prolongation at order 0 is:

```
[[e1, e2, e3, e4, e5, e6, e7], [e8, e9, e10, e11, e12, e13, e14, e15], [e16, e17, e18,
e19, e20, e21, e22, e23, e24, e25, e26, e27, e28, e29, e30, e31, e32, e33, e34, e35, e36,
e37]]
[-2, -1, 0]
```

The Tanaka prolongation at order 1 is:

```
[[e1, e2, e3, e4, e5, e6, e7], [e8, e9, e10, e11, e12, e13, e14, e15], [e16, e17, e18,
e19, e20, e21, e22, e23, e24, e25, e26, e27, e28, e29, e30, e31, e32, e33, e34, e35, e36,
e37], [e38, e39, e40, e41, e42, e43, e44, e45]]
[-2, -1, 0, 1]
```

The Tanaka prolongation at order 2 is:

```
[[e1, e2, e3, e4, e5, e6, e7], [e8, e9, e10, e11, e12, e13, e14, e15], [e16, e17, e18,
e19, e20, e21, e22, e23, e24, e25, e26, e27, e28, e29, e30, e31, e32, e33, e34, e35, e36,
e37], [e38, e39, e40, e41, e42, e43, e44, e45], [e46, e47, e48, e49, e50, e51, e52]]
[-2, -1, 0, 1, 2]
```

We see that the Tanaka prolongation of the symbol algebra terminates at order 2 and has dimension 52. As one might guess, the Tanaka prolongation is also the split real form of  $\mathfrak{f}_4$ .

```
[symalg > DGsetup(TanPr, [t], [alpha]);
```

```
Lie algebra: prsym
```

(4.9)

```
[symalg > ClassifyComplexSemiSimpleLieAlgebra(prsym, properties = 'TanPrProperties');
```

```
Step 1. Find a Cartan subalgebra
```

```
Step 2. Find the root space decomposition
```

```
Step 3. Find the positive and simple roots
```

```
Step 4. Find the Cartan matrix
```

Step 5. Put the Cartan matrix in standard form  
"F4"

(4.10)

#### 4. Associated Parabolic Geometry

The infinitesimal symmetry algebra  $\Gamma$  of our Cartan differential system acts infinitesimally transitively on the underlying manifold  $M$  and therefore  $M$  is locally an  $F_4$  homogeneous space,  $M = F_4/H$ . Since the symbol algebra is a 2-step nilpotent algebra  $\mathfrak{m}(\Delta) = \mathfrak{g}_{-2} \oplus \mathfrak{g}_{-1}$  and the prolongation of the symbol algebra is  $\mathfrak{f}_4$ , one can say substantially more. Namely, the isotropy subalgebra is conjugate to a parabolic subalgebra of  $\mathfrak{f}_4$  associated to a gradation  $\mathfrak{f}_4 = \mathfrak{g}_{-2} \oplus \mathfrak{g}_{-1} \oplus \mathfrak{g}_0 \oplus \mathfrak{g}_1 \oplus \mathfrak{g}_2$ , where  $\dim \mathfrak{g}_{-2} = \dim \mathfrak{g}_2 = 7$ ,  $\dim \mathfrak{g}_{-1} = \dim \mathfrak{g}_1 = 8$  and  $\dim \mathfrak{g}_0 = 12$ . Each gradation of  $\mathfrak{f}_4$  is determined by a subset of the simple roots (see [1], Section 3.2). Since  $\mathfrak{f}_4$  has rank 4, there are 4 possible roots and hence 16 possible gradations. We can use the command [GradeSemiSimpleLieAlgebra](#) to calculate the various gradings of  $\mathfrak{f}_4$ .

We shall use the realization of  $\mathfrak{f}_4$  that we computed from the symbol algebra and the Tanaka prolongation in the previous section with properties stored in the record **TanPrProperties**.

**F4CB > SimRts := TanPrProperties:-SimpleRoots;**

$$SimRts := \left[ \begin{bmatrix} 0 \\ 1 \\ -1 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 0 \\ 1 \\ -2 \end{bmatrix}, \begin{bmatrix} 0 \\ 0 \\ 0 \\ 1 \end{bmatrix}, \begin{bmatrix} \frac{1}{2} \\ -\frac{1}{2} \\ -\frac{1}{2} \\ 0 \end{bmatrix} \right]$$

(5.1)

As is customary, let's label these roots as  $\alpha_1, \alpha_2, \alpha_3, \alpha_4$ .

Here is the gradation of  $\mathfrak{f}_4$  defined by  $\alpha_1$ . It is a 2-step gradation.

```

prsym > Gr1 := GradeSemiSimpleLieAlgebra(SimRts[1..1], TanPrProperties);
Gr1 := table([-1 = [t6, t7, t11, t12, t27, t28, t31, t32, t35, t36, t42, t45, t46, t47], 0 = [t3, t4, t5, t9, t10, t13, t14, t16, t17, t22, t23,
t29, t30, t33, t34, t38, t39, t43, t44, t48, t49, t50], -2 = [t37], 1 = [t1, t2, t8, t15, t18, t19, t20, t24, t25, t26, t40, t41, t51, t52], 2
= [t21]])

```

(5.2)

```

prsym > nops(Gr1[-2]), nops(Gr1[-1]);

```

1, 14

(5.3)

Here is the gradation of  $\mathfrak{f}_4$  defined by  $\alpha_2$ . It is a 3-step gradation.

```

prsym > Gr2 := GradeSemiSimpleLieAlgebra(SimRts[2..2], TanPrProperties);
Gr2 := table([-1 = [t5, t6, t10, t11, t22, t27, t32, t33, t44, t45, t47, t48], 0 = [t4, t9, t13, t16, t23, t24, t28, t29, t34, t38, t43, t49],
-2 = [t7, t12, t31, t36, t42, t46], 1 = [t2, t3, t14, t15, t17, t18, t25, t30, t39, t40, t50, t51], -3 = [t35, t37], 2 = [t1, t8, t19, t26,
t41, t52], 3 = [t20, t21]])

```

(5.4)

```

prsym > nops(Gr2[-3]), nops(Gr2[-2]), nops(Gr2[-1]);

```

2, 6, 12

(5.5)

Here is the gradation of  $\mathfrak{f}_4$  defined by  $\alpha_3$ . It is a 3-step gradation.

```

prsym > Gr3 := GradeSemiSimpleLieAlgebra(SimRts[3..3], TanPrProperties);
Gr3 := table([-1 = [t10, t11, t32, t33, t38, t49], 0 = [t5, t6, t13, t16, t23, t24, t28, t29, t34, t43, t50, t51], -2 = [t7, t12, t22, t27,
t36, t44, t45, t47, t48], 1 = [t4, t9, t25, t30, t39, t40], -3 = [t31, t42], 2 = [t2, t3, t14, t15, t17, t18, t26, t41, t52], -4 = [t35,
t37, t46], 3 = [t8, t19], 4 = [t1, t20, t21]])

```

(5.6)

```

prsym > nops(Gr3[-3]), nops(Gr3[-2]), nops(Gr3[-1]);

```

2, 9, 6

(5.7)

Here is the gradation of  $\mathfrak{f}_4$  defined by  $\alpha_4$ . It is a 2-step gradation. This one gives the correct grading for the symbol algebra and Tanaka prolongation

```

prsym > Gr4 := GradeSemiSimpleLieAlgebra(SimRts[4..4], TanPrProperties);
Gr4 := table([-1 = [t10, t11, t12, t13, t38, t42, t44, t45], 0 = [t2, t3, t4, t5, t6, t16, t23, t24, t25, t26, t28, t29, t30, t32, t33, t34,
t36, t47, t48, t49, t50, t51], -2 = [t7, t22, t27, t31, t35, t37, t46], 1 = [t8, t9, t14, t15, t39, t40, t41, t43], 2 = [t1, t17, t18, t19,

```

(5.8)



```

t20, t21, t52]])
[prsym > nops(Gr4[-2]), nops(Gr4[-1]);
7, 8
(5.9)

```

It is a simple matter to check that no other subsets of roots lead to gradings with the correct length and dimensions. For example, the gradation defined by the roots  $(\alpha_1, \alpha_2)$  is a 5-step gradation.

```

[prsym > Gr12 := GradeSemiSimpleLieAlgebra(SimRts[1..2], TanPrProperties);
Gr12 := table([-1 = [t5, t10, t22, t28, t33, t44, t48], 0 = [t4, t9, t13, t16, t23, t29, t34, t38, t43, t49], -2 = [t6, t11, t27, t32, t45,
t47], 1 = [t3, t14, t17, t24, t30, t39, t50], -3 = [t7, t12, t31, t36, t42, t46], 2 = [t2, t15, t18, t25, t40, t51], -4 = [t35], 3 = [t1,
t8, t19, t26, t41, t52], 4 = [t20], -5 = [t37], 5 = [t21]]);
(5.10)

```

```

[prsym > nops(Gr12[-5]), nops(Gr12[-4]), nops(Gr12[-3]), nops(Gr12[-2]), nops(Gr12[-1]);
1, 1, 6, 6, 7
(5.11)

```

In this way we conclude that the homogeneous space on which our differential system is defined is  $F_4/P$ , where the Lie algebra  $\mathfrak{p}$  of  $P$  is the parabolic subalgebra  $\mathfrak{p} = \mathfrak{g}_0 \oplus \mathfrak{g}_1 \oplus \mathfrak{g}_2$  for the grading  $\mathfrak{f}_4 = \mathfrak{g}_{-2} \oplus \mathfrak{g}_{-1} \oplus \mathfrak{g}_0 \oplus \mathfrak{g}_1 \oplus \mathfrak{g}_2$  defined by the root  $\alpha_4$ . This agrees with the analysis of Yamaguchi [3], page 482.

## Highlighted Commands

[Chevalley basis](#), [InfinitesimalSymmetriesOfEDS](#), [SymbolAlgebra](#), [TanakaProlongation](#)

## References

1. A. Cap and J. Slovak, *Parabolic Geometries I. Background and General Theory. Mathematical Surveys and Monographs*, **154**(2009), American Mathematics Society.
2. E. Cartan, *Über die einfachen Transformationsgruppen*, Leipziger Berichete (1893), 395--420.
3. K. Yamaguchi, *Differential Systems Associated with Simple Lie Algebras*, Advanced Studies in Pure Mathematics, **22**(2009), 413--494.

## Release Notes

- This worksheet was compiled with Maple 17 and DG release USU1, available by request from [ian.anderson@usu.edu](mailto:ian.anderson@usu.edu).

## Authors

Ian. M. Anderson  
Department of Mathematics and Statistics  
Utah State University  
Jan 1, 2015

[ian.anderson@usu.edu](mailto:ian.anderson@usu.edu)